=====StateVariable_S_Z=====




```
*         Y(S)/U(S)  =   K*(1+ s*b1 + s^2*b2)/( s^2 +a2*s +a1)
*         Complex_Poles  1/(s^2 +a2*s +a1)  = 1/(s^2 +(wo/Q)*s +wo^2)
```

If the "1/z" integration block gets scaled to behave like a "1/s" integration block, they both should operate the same way in the same circuit.

```
*=========Create_Signal==================
VT        VT      0       DC      0   PWL( 0 0 100 100)
BVIN      VIN     0       V       =  1*sin(6.283*V(FIN)*V(VT))
BVCLK     CLK     0       V       =  u(sin(6.283*V(FS)*V(VT)))
BF1       F1      0       V       =  2*sin(3.14*V(FC)/V(FS))
VFIN      FIN     0       DC      1
VFC       FC      0       DC      1
VFS       FS      0       DC      50
VK        K       0       DC      1
VA1       A1      0       DC      1
VA2       A2      0       DC      1
VB0       B0      0       DC      1
VB1       B1      0       DC      -1u
VB2       B2      0       DC      1u
XStateS   VIN     FC              K   A1  A2  B0  B1  B2  VOUT  HP  BP  LP  StateVS
XStateZ   VIN     F1      CLK     K   A1  A2  B0  B1  B2  VOUTZ HPZ BPZ LPZ StateVZ
```

The only difference between the "1/z" integration block from the "1/s" integration block is that it needs a scale factor for the clock frequency.

```
.control
*=========Run_At_FIN_eq_1================================
run
set       pensize = 2
plot      vin  bp hp lp bpz hpz lpz   title FreqCutOff_is_1
*=========Run_At_FIN_eq_.7================================
alter     VFIN         dc = .7
run
plot      vin  bp hp lp bpz hpz lpz   title FreqCutOff_is_.7
*=========Run_At_FIN_eq_1.5================================
alter     VFIN         dc = 1.5
run
plot      vin  bp hp lp bpz hpz lpz   title FreqCutOff_is_1.5
.endc
```

Running both types of filters through the same input frequencies should produce the same output results.



At 1Hz, the bandpass for both the s and z version should match closely the input signal. The lowpass and highpass output should be about equal magnitude. The transient response seems to be about the same as well.



At 0.7Hz, the low pass should be higher.



At 1.5Hz, the high pass should be higher. The mapping between the s and z is not perfect. The mapping equations are given below.

$$s = \frac{2}{T}\frac{z-1}{z+1} \qquad z = \frac{2+sT}{2-sT} \qquad z = e^{sT}$$

The responses will not be equal because the non-sampled system does not alias. The jw dimension on a linear system has no limitations.



But on a sampled system, the sample rate sets the nyquist. A very nice mapping between the s and z plane is shown above.

```
===========Full_Netlist_For_Copy_Paste======================
StateVariable_S_Z
*
*                              ____
*                        _____|  b2 |_____________________
*                       |     |_____|     ____            |
*                       |            ____| b1 |________   |
*                       | Highpass  |    |____|        |  |
* U(S)             ..   |           |  .               V_V
*            ___    y(t)|    _____  |  y(t)   _____  y(t) / _ \
*  _\| k |_\/ \_  \_____|_| 1/s |_|______| 1/s |____\/ \   \___\ Y(S)
*   /|___| /\ /_   /       |_____|   |     |_____|    /\ /_  /   /
*           \___/                    |           |      \___/
*            ^  ^         Bandpass   |           |
*            |  |___| -a2 |__________|           |     b0 = 1
*            |      |_____|              Lowpass |
*            |_________________| -a1 |___________|      w0 = sqrt(a1)
*                              |_____|                  Q  = w0/a2
*
*          Y(S)/U(S)   =   K*(1+ s*b1 + s^2*b2)/( s^2 +a2*s +a1)
*          Complex_Poles  1/(s^2 +a2*s +a1)  = 1/(s^2 +(wo/Q)*s +wo^2)
*
*=========Create_Signal==================
VT        VT     0      DC      0   PWL( 0 0 100 100)
BVIN      VIN    0      V       =  1*sin(6.283*V(FIN)*V(VT))
BVCLK     CLK    0      V       =  u(sin(6.283*V(FS)*V(VT)))
BF1       F1     0      V       =  2*sin(3.14*V(FC)/V(FS))
VFIN      FIN    0      DC      1
VFC       FC     0      DC      1
VFS       FS     0      DC      50
VK        K      0      DC      1
VA1       A1     0      DC      1
VA2       A2     0      DC      1
VB0       B0     0      DC      1
VB1       B1     0      DC      -1u
VB2       B2     0      DC      1u
XStateS   VIN    FC          K   A1  A2  B0  B1  B2  VOUT  HP  BP  LP  StateVS
XStateZ   VIN    F1     CLK  K   A1  A2  B0  B1  B2  VOUTZ HPZ BPZ LPZ StateVZ
```

```
*TRAN      TSTEP  TSTOP  TSTART TMAX   ?UIC?
.tran      30u    2      0      30u    UIC
.control
*=========Run_At_FIN_eq_1=================================
run
set        pensize = 2
plot       vin  bp hp lp bpz hpz lpz   title FreqCutOff_is_1
*=========Run_At_FIN_eq_.7=================================
alter      VFIN         dc = .7
run
plot       vin  bp hp lp bpz hpz lpz   title FreqCutOff_is_.7
*=========Run_At_FIN_eq_1.5================================
alter      VFIN         dc = 1.5
run
plot       vin  bp hp lp bpz hpz lpz   title FreqCutOff_is_1.5
.endc
*=========Switch===========================================
.MODEL     SW      SW(     VT=.5 VH=.1  RON=1000m ROFF=100MEG)
*=========StateVariable_Cell_Z=========================================
*
*
* ___     |\  K      HP   |\  b2 gmb2                       ___
*| IN|__| \/\/\________| \/\/\________________________|OUT|
*|___|    | /\/\/        |  /\/\/                        |    |___|
*         |/  gmin       |  |/         |\  b1 gmb1       |
*                        |          __| \/\/\___________|
*                        |         |  | /\/\/
*                        |  Fscale |  |/   Fscale
*          R1 =1         |  __|__  |       __|__   |\  b0   |      R2 =1
*     _/\  /\  /\___|_| 1/z |_|_____| 1/z |___| \/\/\_|_/\  /\  /\_
*    _|_ \/  \/     |  |_____|  |    |_____|  | /\/\/    \/  \/  _|_
*    ///             |  s1       /|        s2  |  |/  gmb0         ///
*                    |______/\/\/ |_|BP        |
*                           \/\/\ |gma1        |
*                           -a2  \|         /| |LP
*                    |________________/\/\/ |_|
*                                     \/\/\ |gma2
*                                     -a1  \|
.SUBCKT   StateVZ VIN    F1  CLK  K   A1  A2  B0  B1  B2  VOUT HP BP LP
 R1        HP     0      1
 R2        OUT    0      1
 Bgmin     HP     0      I  = -V(VIN)*V(K)*1
 Bgma1     HP     0      I  =  V(LP)*V(A1)
 Bgma2     HP     0      I  =  (V(BP))*V(A2)
 Bgmb0     OUT    0      I  = -V(LP)*V(B0)
 Bgmb1     OUT    0      I  = -V(BP)*V(B1)
 Bgmb2     OUT    0      I  = -V(HP)*V(B2)
 XZblock1  HP     BP     F1       CLK  Zblock
 XZblock2  BP     LP     F1       CLK  Zblock
 BOUT      VOUT   0      V  =    V(OUT)
.ENDS      StateVZ
*=========Z_BLOCK==========================================
*
*      F1 = 2*sin(pi*Fc/Fs)
*                ___
* ___     |\ IN1 / _ \                ___
*|VIN|__|1\__\/ \  2\__________|OUT|
*|___|    | /  /\ /_  /          |    |___|
*         |/     \___/  | CLK    |
*           IN2   ^    __V_____  |
*                 |__| Z_DLAY |_|
*                      |_______|
*
.SUBCKT   Zblock VINZ   OUTZ   F1   CLK
 B1       IN1    0      V =    V(VINZ)*V(F1)
 B2       OUTZ   0      V =    V(IN1) + V(IN2)
 XZ_DLAY  OUTZ   IN2    CLK    Z_DLAY
.ENDS     Zblock
*=========Z_DLAY========================================
*                           _________________
*                          |    ____________  |
*   ___    |\         VBF  |   |    ____    | |  ____
*  |CLK|__|1\______________|_/ |\_\     \   | |_|    \
*  |___|   | /  |        VLP   |   |NE \_   |   | PE  \_
*          |/   |/\  /\  /\_____|___| 2 /    |___| 3   /
*                  \/  \/   _|_      /___/ |      |___/ |
*                  RLP      ___                         |
*                            _|_
*                  CLP ///
*
*        |\  V1  /   V2         |\  V3  /   V4      |\
*|IN |__|4\____/ ___________|5\____/ __________|6\_|OUT|
*|___|   | / S1  NE   |  _|_    | / S2  PE   |  _|_  | /  |___|
*        |/           |  ___    |/           |  ___  |/
```

```
*        Rshunt1    |  | C1   Rshunt2    |  |C2
*      _/\  /\  /\__| _|_   _/\  /\  /\__| _|_
*    _|_  \/  \/      /// _|_  \/  \/      ///
*    ///                  ///
.SUBCKT   Z_DLAY VIN    OUT    CLK
 B1       VBF    0      V =    u( v(CLK )-.5 )
 RLP      VBF    VLP    10k
 CLP      VLP    0      50n    IC=0
 BNOR2    NE     0      V =    1-u( u(v(VBF )-.5)+u(.5 -v(VLP ) ) -.1)
 BAND3    PE     0      V =      u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.1)
 B4       V1     0      V =    V(VIN)
 S1       V1     V2     NE     0    SW
 R1       V2     0      100Meg
 C1       V2     0      1u
 B5       V3     0      V =    V(V2)
 S2       V3     V4     PE     0    SW
 R2       V4     0      100Meg
 C2       V4     0      100u
 B6       OUT    0      V =    V(V4)
.ENDS     Z_DLAY
*=========StateVariable_Cell_S===========================================
*
*
* ___   |\  K      HP   |\  b2 gmb2                       ___
*| IN|__| \/\/\_________| \/\/\__________________________|OUT|
*|___|  | /\/\/     |   | /\/\/                        | |___|
*       |/  gmin    |   |/       |\  b1 gmb1           |
*                   |          __| \/\/\_______________|
*                   |          | | /\/\/               |
*                   |  Fscale  | |/   Fscale           |
*          R1 =1    |  ___|__  |      __|__    |\  b0  |     R2 =1
*     _/\  /\  /\___|_| 1/s  |_|_____| 1/s |___| \/\/\_|_/\  /\  /\_
*    _|_ \/  \/   |   |______| | |   |_____||  | /\/\/     \/  \/ _|_
*    ///          | s1        /| |      s2  |  |/  gmb0           ///
*                 |______/\/\/ |_|BP        |
*                 |      \/\/\ |gma1        |
*                 |      -a2  \|         /| |LP
*                 |_________________/\/\/ |_|
*                                   \/\/\ |gma2
*                                   -a1  \|
.SUBCKT   StateVS VIN    FC   K   A1  A2  B0  B1  B2  VOUT  HP  BP  LP
 R1        HP     0      1
 R2        OUT    0      1
 Bgmin     HP     0      I  = -V(VIN)*V(K)*1
 Bgma1     HP     0      I  =  V(LP)*V(A1)
 Bgma2     HP     0      I  =  (V(BP))*V(A2)
 Bgmb0     OUT    0      I  = -V(LP)*V(B0)
 Bgmb1     OUT    0      I  = -V(BP)*V(B1)
 Bgmb2     OUT    0      I  = -V(HP)*V(B2)
 XS1block  HP     BP     FC     Sblock
 XS2block  BP     LP     FC     Sblock
 BOUT      VOUT   0      V  =   V(OUT)
.ENDS     StateVS
*=========S_BLOCK===========================================
*                          _______
*         IN   RS=1/FC   _|_      |
* ___   |\               ___      |
*|VIN|__| \_/\  /\  /\____|___|\   |  ___
*|___|  | /   \/  \/   VNEG   | \_|_|OUT|
*       |/                ____| /   |___|
*                        _|_  |/
*   For FC = 1Hz         ///
*   RS = 1 Ohm                CS=1uF/2Pi
*   Xc = 1 Ohm
.SUBCKT   Sblock VIN    OUT    FC
 Bbuf     IN     0      V =    -V(VIN)
 BRS      IN     VNEG   I =    (V(IN)-V(VNEG))/V(FC)
 Cs       VNEG   OUT    .159
 BSOUT    OUT    0      V =    -V(VNEG)*1000
.ENDS     Sblock
.end
```

6.7.11_11.01AM

dsauersanjose@aol.com
Don Sauer